


Application Note		資料作成 100428	資料番号 an2k-020
テーマ	タッチパネルのトレンドグラフを使ってみる		
使用機器	MPC-1000、タッチパネル(キーエンス VT4-W4G)、CHINO 温湿度計		

■機器構成




■概要

- ・トレンドグラフとは、大雑把にいうと、タッチパネルが起動時から一定間隔で、特定箇所のデータを読み、グラフ上にプロットしていく、というものです(正確にはタッチパネルの取説をご覧ください)。
- ・つまり、MPC は他の動作と関係なくバックグラウンドで通信と書き込みをしていれば、タッチパネルが自動的にグラフを描いてくれるわけです。
- ・(タッチパネル機種により違いがあるようですが)トレンドグラフのデータをメモリカードなどに保存ができるようで、簡単にロガーを作れそうです。
- ・このサンプルでは、セレクト SW でトレンドグラフ画面と棒グラフ画面を切り換えます。

■画面デザイン

- ・トレンドグラフ画面(画面番号 136)



文字表示(開始時刻:DT00401)
 文字表示(温度 CH1:DT00135、CH2:DT00145)
 文字表示(湿度 CH1:DT00140、CH2:DT00150)

トレンドグラフ ID0


グラフの種類：リアルタイム
 グラフ本数：2、表示データ数 240
 入力レンジ：下限値 0、上限値 40
 項目軸目盛：目盛割付 等分割、分割数 12
 数値軸目盛：目盛割付 等分割、分割数 4
 トレンドサンプルデバイス：DT00400~DT00401
 データ長：1 ワード
 サンプリングトリガ 内部タイマ(自動)
 サンプリング周期：180 秒

トレンドグラフ ID1

グラフの種類：リアルタイム
 グラフ本数：2、表示データ数 240
 入力レンジ：下限値 0、上限値 100
 項目軸目盛：目盛割付 等分割、分割数 12
 数値軸目盛：目盛割付 等分割、分割数 4
 トレンドサンプルデバイス：DT00402~DT00403
 データ長：1 ワード
 サンプリングトリガ 内部タイマ(自動)
 サンプリング周期：180 秒

- ・棒グラフ画面(画面番号 137)

文字表示 (温度 DT00135,湿度 DT00140,温度 DT00145,湿度 DT00150)



文字表示 (年月日 DT00124 10 桁)

文字表示 (時分秒 DT00116 8 桁)

数値表示(通信トータル回数 DT00162)

数値表示(CH1 エラー回数 DT00160)

数値表示(CH2 エラー回数 DT00161)

バー形メータ(温度 DT00400,湿度 DT00401,温度 DT00402,湿度 DT00403)

■MPC プログラム

午前 9 時前に電源を入れると、09:00 にグラフをトレンドグラフリセットし開始時刻とします。

```
*_BEGIN
CONST PAGING 4 /* タッチパネル画面切り替えアドレス
CONST GRAPH_START &H090000 /*トレンドグラフリセット時間 時分秒

MEWNET 38400 2 /* タッチパネル ボーレート、通信ポート
S_MBK 99 PAGING /* オープニング画面
CNFG# 1 RS485 "9600b7pes1NONE" /* 温湿度計通信設定
SETIO
TIME 1000 /* need for VT start
GOSUB *CLOCK_SET /* タッチパネルの時計で MPC の時計をアジャスト

QUIT 1
FORK 1 *MODE /* セレクト SW で画面を切り替えるタスク
QUIT 10
FORK 10 *TEMP_HUMI /* 温湿度計通信と時計表示

END /* メインタスク終了

*MODE /* セレクト SW で画面を切り替えるタスク
DO
IF SW(195)==0 THEN /* セレクト SW
S_MBK 136 PAGING /*トレンドグラフ画面
WAIT SW(195)==1 /* セレクト SW
ELSE /* MANU
S_MBK 137 PAGING /* 棒グラフ画面
WAIT SW(195)==0 /* セレクト SW
END_IF
LOOP

*CLOCK_SET /* タッチパネル時計 -> MPC 時計
/* キーエンス VT3-W4G
/* MBK(0) = 00yy [ 年]
/* MBK(1) = 00mm [ 月]
/* MBK(2) = 00dd [ 日]
/* MBK(3) = hhmm [時分]

FORMAT ""
TODAY=&H20000000+MBK(0)*&H10000
TODAY=TODAY+(MBK(1)*&H100)+MBK(2)
NOW=MBK(3)*&H100
PRINT "VT CLOCK" HEX$(TODAY) HEX$(NOW)
SET_RTC TODAY
SET_RTC NOW
RETURN

*TEMP_HUMI /* 温湿度計通信と時計表示
GOSUB *TEMP_GRAPH_RESET
DO
FORMAT "0000/00/00"
DD$=HEX$(DATE(0))
S_MBK DD$ 124 10 /* タッチパネル 年月日表示

FORMAT "00:00:00"
TT$=HEX$(TIME(0))
OLD_TIME=TIME(0)
S_MBK TT$ 116 8 /* タッチパネル 時間表示
PRINT DD$ TT$

IF (TIME(0)/&H10)==(GRAPH_START/&H10) THEN /* 指定時間でグラフリセット(秒1位切り捨て)
TIME 10000
GOSUB *TEMP_GRAPH_RESET
END_IF

IF CH%2==0 THEN /* 温湿度計の反応が遅いので交互に取得
GOSUB *TEMP_HUMI_GET 1 135 140 400 402
TEMP1$=TEMP$
HUMI1$=HUMI$
ELSE
GOSUB *TEMP_HUMI_GET 2 145 150 401 403
TEMP2$=TEMP$
HUMI2$=HUMI$
END_IF
CH=CH+1
S_MBK CH 162
WAIT OLD_TIME<>TIME(0)

LOOP

*TEMP_HUMI_GET /* 温湿度取得
```

```

_VAR CH_NO MBK_T MBK_H MBK_TG MBK_HG
FORMAT ""
SEND$=CHR$(5) /* ENQ
SEND$=SEND$+"0"+STR$(CH_NO) /* 機器番号
SEND$=SEND$+CHR$(2) /* STX
SEND$=SEND$+"R" /* 読み出し
SEND$=SEND$+"PV" /* 測定データ
SEND$=SEND$+"01" /* データ番号
SEND$=SEND$+CHR$(3) /* ETX
SEND$=SEND$+CHR$(13)+CHR$(10)

INPUT# 1 CLR_BUF
PRINT# 1 SEND$
RCV$=""
DO
  INPUT# 1 CHR_C|1 TMOUT|2 T$
  IF T$==CHR$(&HA) OR rse_==1 THEN
    BREAK
  END_IF
  RCV$=RCV$+T$
LOOP

IF rse ==0 THEN
  STRCPY RCV$ TEMP$ 35 4
  STRCPY RCV$ HUMI$ 44 4
  S_MBK TEMP$ MBK_T 4 /* タッチパネル 温度 文字列
  S_MBK HUMI$ MBK_H 4 /* タッチパネル 湿度 文字列

  S_MBK VAL(TEMP$) MBK_TG /* タッチパネル 温度 メータ
  S_MBK VAL(HUMI$) MBK_HG /* タッチパネル 湿度 メータ

  S_MBK ERR_CNT1 160
  S_MBK ERR_CNT2 161
  PRINT CH_NO TEMP$ HUMI$
ELSE
  IF CH_NO==1 THEN
    ERR_CNT1=ERR_CNT1+1
    PRINT "ERR1 " ERR_CNT1
  ELSE
    ERR_CNT2=ERR_CNT2+1
    PRINT "ERR2 " ERR_CNT2
  END_IF
  TEMP$=" ER"+STR$(rse_)
  HUMI$=""
  S_MBK TEMP$ MBK_T 4
  S_MBK HUMI$ MBK_H 4
END_IF

RETURN


*TEMP_GRAPH_RESET
ON 75000 /* グローバル機能制御 トレンドグラフ制御 データクリア
WAIT SW(75000)==0
FORMAT "00:00:00"
TT$=HEX$(TIME(0))
S_MBK TT$ 404 8
CH=0
ERR_CNT1=0
ERR_CNT2=0

PRINT "GRAPH RESET"


RETURN

```

■ 実働画面



トレンドグラフ画面



棒グラフ画面